Journal of Software for Algebra and Geometry

Partially ordered sets in Macaulay2

David Cook II, Sonja Mapes and Gwyneth Whieldon

vol 7 2015

JSAG **7** (2015), 9–15
dx.doi.org/10.2140/jsag.2015.7.9

The Journal of Software for
Algebra and Geometry

Partially ordered sets in Macaulay2

DAVID COOK II, SONJA MAPES AND GWYNETH WHIELDON

ABSTRACT: We introduce the package *Posets* for Macaulay2. This package provides a data structure and the necessary methods for working with partially ordered sets, also called posets. In particular, the package implements methods to enumerate many commonly studied classes of posets, perform operations on posets, and calculate various invariants associated to posets.

INTRODUCTION. A *partial order* is a binary relation $\preceq$ over a set P that is antisymmetric, reflexive, and transitive. A set P together with a partial order $\preceq$ is called a *poset*, or *partially ordered set*. We refer the reader to the seminal text [Stanley 2012] for definitions omitted herein.

Posets are combinatorial structures that are used in modern mathematical research, particularly in algebra. We introduce the package *Posets* for Macaulay2 [Grayson and Stillman] via three distinct posets or related ideals which arise naturally in combinatorial algebra.

We first describe two posets that are generated from algebraic objects. The intersection semilattice associated to a hyperplane arrangement can be used to compute the number of unbounded and bounded real regions cut out by a hyperplane arrangement, as well as the dimensions of the homologies of the complex complement of a hyperplane arrangement.

Given a monomial ideal, the lcm-lattice of its minimal generators gives information on the structure of the free resolution of the original ideal. Specifically, two monomial ideals with isomorphic lcm-lattices have the "same" (up to relabeling) minimal free resolution, and the lcm-lattice can be used to compute, among other things, the multigraded Betti numbers

$$\beta_{i,\boldsymbol{b}}(R/M) = \dim_K \operatorname{Tor}_{i,\boldsymbol{b}}(R/M, K)$$

of the monomial ideal.

In contrast to the first two examples (associating a poset to an algebraic object), we then describe an ideal that is generated from a poset. In particular, the Hibi ideal of a finite poset is a squarefree monomial ideal which has many nice *algebraic*

MSC2010: 06A11, 13P99.
Keywords: posets.



Figure 1. Left: the noncentral hyperplane arrangement $\mathcal{A}$ comprising $\mathcal{H}_1 = V(x+y)$, $\mathcal{H}_2 = V(x)$, $\mathcal{H}_3 = V(x-y)$, $\mathcal{H}_4 = V(y+1)$. Right: The intersection meet semilattice $\mathcal{L}(\mathcal{A})$ of $\mathcal{A}$.

properties that can be described in terms of *combinatorial* properties of the poset. For example, the minimal free resolution, the Betti numbers, and the projective dimension are nicely described in terms of data about the poset itself.

INTERSECTION (MEET SEMI)LATTICES. A *hyperplane arrangement* $\mathcal{A}$ is a finite collection of affine hyperplanes in some vector space V. The *dimension* of a hyperplane arrangement is defined by $\dim(\mathcal{A}) = \dim(V)$, and the *rank* of a hyperplane arrangement $\operatorname{rank}(\mathcal{A})$ is the dimension of the span in V of the set of normals to the hyperplanes in A.

The *intersection (meet semi)lattice* $\mathcal{L}(\mathcal{A})$ of $\mathcal{A}$ is the set of the nonempty intersections of subsets of hyperplanes $\bigcap_{\mathcal{H}\in\mathcal{A}'} \mathcal{H}$ for $\mathcal{H} \in \mathcal{A}' \subseteq \mathcal{A}$, ordered by reverse inclusion. We include the empty intersection corresponding to $\mathcal{A}' = \varnothing$, which is the minimal element $\hat{0} = V$ in the intersection meet semilattice $\mathcal{L}(A)$. If the intersection of all hyperplanes in $\mathcal{A}$ is nonempty, i.e., $\bigcap_{\mathcal{H}\in\mathcal{A}} \mathcal{H} \neq \varnothing$, then the intersection meet semilattice $\mathcal{L}(A)$ is actually a lattice. Arrangements with this property are called *central arrangements*.

Consider the noncentral hyperplane arrangement

$$\mathcal{A} = \{\mathcal{H}_1 = V(x+y), \mathcal{H}_2 = V(x), \mathcal{H}_3 = V(x-y), \mathcal{H}_4 = V(y+1)\},$$

where $\mathcal{H}_i = V(\ell_i(x, y)) \subseteq \mathbb{R}^2$ denotes the hyperplane of zeros of the linear form $\ell_i(x, y)$; see Figure 1, left. We construct $\mathcal{L}(\mathcal{A})$ in Macaulay2 as follows.

```
i1 : needsPackage "Posets";
i2 : R = RR[x,y];
i3 : A = {x + y, x, x - y, y + 1};
i4 : LA = intersectionLattice(A, R);
```

Further, using the method `texPoset` we generate LaTeX code to display the Hasse diagram of $\mathcal{L}(\mathcal{A})$, as in Figure 1, right.

A theorem of Zaslavsky [1975] provides information about the topology of the complement of hyperplane arrangements in $\mathbb{R}^n$. Let μ denote the Möbius function of the intersection meet semilattice $\mathcal{L}(\mathcal{A})$. Then the number of regions that $\mathcal{A}$ divides $\mathbb{R}^n$ into is

$$r(\mathcal{A}) = \sum_{x \in \mathcal{L}(\mathcal{A})} |\mu(\hat{0}, x)|.$$

Moreover, the number of these regions that are bounded is

$$b(\mathcal{A}) = |\mu(\mathcal{L}(\mathcal{A}) \cup \hat{1})|,$$

where $\mathcal{L}(\mathcal{A}) \cup \hat{1}$ is the intersection meet semilattice adjoined with a maximal element.

We verify these results for the noncentral hyperplane arrangement $\mathcal{A}$ using Macaulay2:

```
i5 : realRegions(A, R)
o5 = 10
i6 : boundedRegions(A, R)
o6 = 2
```

Moreover, in the case of hyperplane arrangements in $\mathbb{C}^n$, using a theorem of Orlik and Solomon [1980] we can recover the Betti numbers (dimensions of homologies) of the complement $\mathcal{M}_{\mathcal{A}} = \mathbb{C}^n - \bigcup \mathcal{A}$ of the hyperplane arrangement using purely combinatorial data of the intersection meet semilattice. In particular, $\mathcal{M}_{\mathcal{A}}$ has torsion-free integral homology with Betti numbers given by

$$\beta_i(\mathcal{M}_{\mathcal{A}}) = \dim_{\mathbb{C}}(\mathrm{H}_i(\mathcal{M}_{\mathcal{A}})) = \sum_{\substack{x \in \mathcal{L}(\mathcal{A}) \\ \dim_{\mathbb{C}}(x) = n-i}} |\mu(\hat{0}, x)|,$$

where $\mu(\,\cdot\,)$ again represents the Möbius function. See [Wachs 2007] for details and generalizations of this formula.

Posets will compute the ranks of elements in a poset, where the ranks in the intersection meet semilattice `LA` are determined by the codimension of elements. Combining the outputs of our rank function with the Möbius function allows us to calculate that $\beta_0(\mathcal{M}_{\mathcal{A}}) = 1$, $\beta_1(\mathcal{M}_{\mathcal{A}}) = 4$, and $\beta_2(\mathcal{M}_{\mathcal{A}}) = 5$:

```
i7 : RLA = rank LA
o7 =  {{ideal 0}, {ideal(x+y), ideal(x), ideal(x-y), ideal(y+1)},
        {ideal(y,x), ideal(y+1,x-1),ideal(y+1,x), ideal(y+1,x+1)}}
i8 : MF = moebiusFunction LA;
i9 : apply(RLA, r -> sum(r, x -> abs MF#(ideal 0_R, x)))
o9 = {1, 4, 5}
```

LCM-LATTICES. Let $R = K[x_1, \dots, x_t]$ be the polynomial ring in t variables over the field K, where the degree of x_i is the standard basis vector $e_i \in \mathbb{Z}^t$. Let $M = (m_1, \dots, m_n)$ be a monomial ideal in R; then we define the *lcm-lattice* of M,



Figure 2. The lcm-lattice for $M = (a^3b^2c, a^3b^2d, a^2cd, abc^2d, b^2c^2d)$.

denoted by L_M, to be the set of all least common multiples of subsets of the generators of M, partially ordered by divisibility. It is easy to see that L_M will always be a finite atomic lattice. While lcm-lattices are nicely structured, they can be difficult to compute by hand, especially for large examples or for ideals where L_M is not ranked.

Consider the ideal

$$M = (a^3b^2c, a^3b^2d, a^2cd, abc^2d, b^2c^2d)$$

in $R = K[a, b, c, d]$. We construct L_M in Macaulay2 as follows. See Figure 2 for the Hasse diagram of L_M, as generated by the `texPoset` method.

```
i10 : R = QQ[a,b,c,d];
i11 : M = ideal(a^3*b^2*c, a^3*b^2*d, a^2*c*d, a*b*c^2*d, b^2*c^2*d);
i12 : LM = lcmLattice M;
```

Lcm-lattices, which were introduced by Gasharov, Peeva, and Welker [Gasharov et al. 1999], have become an important tool used in studying free resolutions of monomial ideals. There have been a number of results that use the lcm-lattice to give constructive methods for finding free resolutions for monomial ideals; for some examples see [Clark 2010; Peeva and Velasco 2011; Velasco 2008].

In particular, Gasharov, Peeva, and Welker [Gasharov et al. 1999] provided a key connection between the lcm-lattice of a monomial ideal M of R and its minimal free resolution; namely, one can compute the (multigraded) Betti numbers of R/M using the lcm-lattice. Let $\Delta(P)$ denote the order complex of the poset P; then for $i \geq 1$ we have

$$\beta_{i,b}(R/M) = \dim \widetilde{H}_{i-2}(\Delta(\hat{0}, b); K)$$

for all $b \in L_M$, and so

$$\beta_i(R/M) = \sum_{b \in L_M} \dim \widetilde{H}_{i-2}(\Delta(\hat{0}, b); K).$$

These computations can all be done using *Posets* together with the package *SimplicialComplexes* by S. Popescu, G. Smith, and M. Stillman. In particular, we show that $\beta_{i,a^2b^2c^2d} = 0$ for all i with the following calculation:

```
i13 : D1 = orderComplex(openInterval(LM, 1_R, a^2*b^2*c^2*d));
i14 : prune HH(D1)
o14 = -1 : 0
       0 : 0
       1 : 0
o14 : GradedModule
```

Similarly, we show that $\beta_{1,a^3b^2cd} = 2$:

```
i15 : D2 = orderComplex(openInterval(LM, 1_R, a^3*b^2*c*d));
i16 : prune HH(D2)
o16 = -1 : 0
             2
       0 : QQ
o16 : GradedModule
```

HIBI IDEALS. Let $P = \{p_1, \ldots, p_n\}$ be a finite poset with partial order $\preceq$, and let K be a field. The *Hibi ideal*, introduced by Herzog and Hibi [2005], of P over K is the squarefree ideal H_P in $R = K[x_1, \ldots, x_n, y_1, \ldots, y_n]$ generated by the monomials

$$u_I := \prod_{p_i \in I} x_i \prod_{p_i \notin I} y_i,$$

where I is an order ideal of P, i.e., for every $i \in I$ and $p \in P$, if $p \preceq i$ then $p \in I$.

NB: The Hibi ideal is the ideal of the monomial generators of the Hibi ring, a toric ring first described by Hibi [1987].

```
i17 : P = divisorPoset 12;
i18 : HP = hibiIdeal P
o18 = monomialIdeal (x x x x x x , x x x x x y , x x x x y y ,
                      0 1 2 3 4 5   0 1 2 3 4 5   0 1 2 4 3 5
       x x x x y y , x x x y y y , x x x y y y , x x y y y y ,
        0 1 2 3 4 5   0 1 3 2 4 5   0 1 2 3 4 5   0 2 1 3 4 5
       x x y y y y , x y y y y y , y y y y y y )
        0 1 2 3 4 5   0 1 2 3 4 5   0 1 2 3 4 5
```

Herzog and Hibi [2005] proved that every power of H_P has a linear resolution, and the i-th Betti number $\beta_i(R/H_P)$ is the number of intervals of the distributive lattice $\mathcal{L}(P)$ of P isomorphic to the rank-i boolean lattice. Using [Stanley 2012, Exercise 3.47], we recover this by looking instead at the number of elements of $\mathcal{L}(P)$ that cover exactly i elements:

```
i19 : betti res HP
              0  1  2 3

o19 = total: 1 10 12 3
          0: 1  .  . .
          1: .  .  . .
          2: .  .  . .
          3: .  .  . .
          4: .  .  . .
          5: . 10 12 3

i20 : LP = distributiveLattice P;
i21 : cvrs = partition(last, coveringRelations LP);
i22 : -- Determine the number of elements each element covers.
      iCvrs = tally apply(keys cvrs, i -> #cvrs#i);
i23 : -- Turn iCvrs into a list indexed by integers.
      gk = prepend(1, apply(sort keys iCvrs, k -> iCvrs#k))
o23 : {1, 6, 3}
i24 : -- Determine the number of intervals of LP isomorphic
      -- to boolean lattices of a given rank.
      apply(#gk, i -> sum(i..<#gk, j -> binomial(j, i) * gk_j))
o24 : {10, 12, 3}
```

Moreover, Herzog and Hibi [2005] proved that the projective dimension of H_P is the Dilworth number of P, i.e., the maximum length of an antichain of P.

```
i25 : pdim module HP == dilworthNumber P
o25 = true
```

REFERENCES.

[Clark 2010] T. B. P. Clark, "Poset resolutions and lattice-linear monomial ideals", *J. Algebra* **323**:4 (2010), 899–919. MR 2011b:13037 Zbl 1203.13016

[Gasharov et al. 1999] V. Gasharov, I. Peeva, and V. Welker, "The lcm-lattice in monomial resolutions", *Math. Res. Lett.* **6**:5-6 (1999), 521–532. MR 2001e:13018 Zbl 0970.13004

[Grayson and Stillman] D. R. Grayson and M. E. Stillman, "Macaulay2: a software system for research in algebraic geometry", Available at http://www.math.uiuc.edu/Macaulay2.

[Herzog and Hibi 2005] J. Herzog and T. Hibi, "Distributive lattices, bipartite graphs and Alexander duality", *J. Algebraic Combin.* **22**:3 (2005), 289–302. MR 2006h:06004 Zbl 1090.13017

[Hibi 1987] T. Hibi, "Distributive lattices, affine semigroup rings and algebras with straightening laws", pp. 93–109 in *Commutative algebra and combinatorics* (Kyoto, 1985), edited by M. Nagata and H. Matsumura, Adv. Stud. Pure Math. **11**, North-Holland, Amsterdam, 1987. MR 90b:13024 Zbl 0654.13015

[Orlik and Solomon 1980] P. Orlik and L. Solomon, "Combinatorics and topology of complements of hyperplanes", *Inventiones Math.* **56**:2 (1980), 167–189. MR 81e:32015 Zbl 0432.14016

[Peeva and Velasco 2011] I. Peeva and M. Velasco, "Frames and degenerations of monomial resolutions", *Trans. Amer. Math. Soc.* **363**:4 (2011), 2029–2046. MR 2011k:13021 Zbl 1221.13024

[Stanley 2012] R. P. Stanley, *Enumerative combinatorics, I*, 2nd ed., Cambridge Studies in Advanced Mathematics **49**, Cambridge University Press, 2012. MR 2868112 Zbl 1247.05003

[Velasco 2008] M. Velasco, "Minimal free resolutions that are not supported by a CW-complex", *J. Algebra* **319**:1 (2008), 102–114. MR 2008j:13028 Zbl 1133.13015

[Wachs 2007] M. L. Wachs, “Poset topology: tools and applications”, pp. 497–615 in *Geometric combinatorics*, edited by E. Miller et al., IAS/Park City Math. Ser. **13**, Amer. Math. Soc., Providence, RI, 2007. MR 2383132 Zbl 1135.06001

[Zaslavsky 1975] T. Zaslavsky, *Facing up to arrangements: face-count formulas for partitions of space by hyperplanes*, Mem. Amer. Math. Soc. **154**, Amer. Math. Soc., Providence, RI, 1975. MR 50 #9603 Zbl 0296.50010

RECEIVED: 21 Jan 2014 REVISED: 14 Aug 2014 ACCEPTED: 5 Jun 2015

DAVID COOK II:

dwcook@eiu.edu
Department of Mathematics & Computer Science, Eastern Illinois University, Charleston, IN 46613, United States

SONJA MAPES:

smapes1@nd.edu
Department of Mathematics, University of Notre Dame, Notre Dame, IN 46556, United States

GWYNETH WHIELDON:

whieldon@hood.edu
Department of Mathematics, Hood College, Frederick, MD 21701, United States

JOURNAL OF SOFTWARE FOR ALGEBRA AND GEOMETRY vol 7, no 1, 2015

Software for multiplier ideals 1
Zach Teitler

Partially ordered sets in Macaulay2 9
David Cook II, Sonja Mapes and Gwyneth Whieldon

Free resolutions and modules with a semisimple Lie group action 17
Federico Galetto

Computing characteristic classes and the topological Euler characteristic of complex projective schemes 31
Christine Jost

1948-7916(2015)7:1;1-J